
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 6.2.3.3 which is part of the POS test suite ‘POS_14wk10’ in the ‘iwd-EUTRA-B2013-03_D14wk10’ delivery.
The test case can be demonstrated to run with one WCDMA UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_6_2_3_3
Test Group:
POS\6
ATS Version:
iwd-EUTRA-B2013-03_D14wk10
System Simulator used:
R&S CMW500  
UE used:
Nvidia T4i 
Verification Status:
PASS


4. Corrections required for test case 6.2.3.3
Change 1 – Correction to function ‘f_TC_6_2_3_3_UTRAN’
	Function name
	f_TC_6_2_3_3_UTRAN()
+ imports

	Reason for change
	1.   Setting of the UTRAN cell power and switching on the UE is missing in the TC preamble, thus preventing to enter the TC body.

Similarily, steps for switch off the UE are missing in a postamble part after the TC body. 
2.   Steps 5 and 6 can also happen in reverse order, respectively the result of the TC is influenced by the latency of the test environment. E.g. while the user confirms that the UE displayed the notification, the Release Complete message is already sent by the UE, thus causing the TC to FAIL as the message was not expected at that point. . 

	Summary of change
	1.   Added function calls f_SetCellPower() and f_UT_SwitchOnUE() to the TC preamble. The latter required to add an import from UpperTesterCommon to the Imports declaration of module UTRAN_AGNSS_Testcases.ttcn.

Steps for UE switch off need to be added by MCC-TF160. 
2.   Interleaving was used to guarantee that steps 5 and 6 are fulfilled, while the order is irrelevant. 

	TTCN module
	POS\6\UTRAN_AGNSS_Testcases.ttcn

	MCC160 Comment
	


Before change

...

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2014, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: IWD_14wk10
// $Date: 2014-03-06 15:49:52 +0100 (Thu, 06 Mar 2014) $
// $Rev: 10973 $
/******************************************************************************/
module UTRAN_AGNSS_Testcases
{
//=========================================================================
// Imports
//=========================================================================
  import from UTRAN_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from UTRAN_ASP_Definitions language "ASN.1:2002" all;
  import from LCS_Definitions_Arguments language "ASN.1:2002" all with {encode "BER_EncRule"};
  import from TCAP_Definitions language "ASN.1:2002" all with {encode "BER_EncRule"}; //@sic R5-140685 sic@
  import from TCAP_Definitions_Para language "ASN.1:2002" all with {encode "DIRECT_EncRule"};
  import from UTRAN_Component all;
  import from UTRAN_CellInfo all;
  import from UTRAN_POS_Component all;
  import from UTRAN_ConfigurationSteps all;
  import from UTRAN_CommonFunctions all;
  import from UTRAN_TypeDefs all;
  import from UTRAN_CommonDefs all;
  import from UTRAN_AGNSS_Templates all;
  import from UpperTesterFunctions all;
  import from Parameters all;
  import from UTRAN_TTCN_ASP_Templates all;
  import from UTRAN_NAS_PDU_Templates all;
  import from NAS_24008Templates_IRAT all;
  import from UTRAN_CommonProcedures all;
  import from NAS_CommonTemplates all;
  import from UTRAN_AGNSS_Functions all;
  import from POS_Parameters all;
  import from CommonDefs all;
  import from PosSystem_CommonUTRAN all;
  import from ComponentsType all; //@sic R5-140685 sic@
  import from LCS_Objects_Templates all;
  import from PosSystem_Common all;
  import from SS_EncDecComponents all;
  import from NAS_CommonTypeDefs all;
  import from UTRAN_SS_CommonTemplates all; //@sic R5-140685 sic@
  import from SS_CommonTemplates all; //@sic R5-140685 sic@
//==============================================================================
// NI-LR Emergency Call: UE-Based A-GNSS
//==============================================================================
...

//==============================================================================
// Location Notification
//==============================================================================
   /*
   * @desc      REFERENCE TS 37.571-2 clause 6.2.3.3
   * @status    NOT APPROVED
   */
   function f_TC_6_2_3_3_UTRAN() runs on UTRAN_PTC
   {
    var START_Value v_START_Value := '00000000000000000000'B;
    var RRC_DATA_IND v_RRC_DataInd;
    var TI v_Ti_S;
    var octetstring v_Octetstring;
    var octetstring v_EncodedComponents;
    var Components v_Components;
    f_UTRAN_Init(UTRAN_Only);
    f_UTRAN_SS_CreateCellDCH (utran_Cell5);
    f_UTRAN_SendDefSysInfo (utran_Cell5);
    f_UTRAN_IdleUpdated (utran_Cell5);
    f_UTRAN_PageCS(utran_Cell5, terminatingConversationalCall, oct2bit(px_TMSI_Def));
    f_UTRAN_RRC_ConnEst (utran_Cell5, terminatingConversationalCall);
    U_Dc.receive (car_InitDirectTransfer (utran_CellDedicated,
                                          tsc_RB3,
                                          cr_U_PagRsp (?, cr_MobileIdAny(omit))))-> value v_RRC_DataInd;
    v_START_Value := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value);
    f_UTRAN_TestBody_Set(true);
    // Steps 1 and 2. MM Authentication
    f_UTRAN_MM_Authentication ();
    // Step 3. Security Mode procedure
    f_UTRAN_RRC_Security(utran_Cell5, true, cs_domain);
    // Step 4. Send REGISTER
    v_EncodedComponents := f_SS_EncodeComponents (cs_Components (cs_InvokeComponent (cs_InvokeLocNotification (tsc_InvokeId,
                                                                                                            notifyLocationAllowed,
                                                                                                            f_POS_Get_LcsClientAddress (px_LcsClientAddressTOA,
                                                                                                                                    px_LcsClientAddressNPI,
                                                                                                                                    px_LcsClientAddressDigits),
                                                                                                            f_CharPacking_IA5toGsm7Bit (px_LcsClientName, USSD_Packing)))));
    U_Dc.send (cas_DataReq_CS (utran_CellDedicated,
                               tsc_RB3,
                               cs_U_RegisterMT (cs_TI_MT, cs_FacilityIE_tlv(v_EncodedComponents))));
    // Step 5. UE notifies the UE user of the location request
    f_UT_MMI_CheckDisplayLcsClient(UT);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.2.3.3 Step 5");
    v_Ti_S := valueof(cs_TI_MT);
    v_Ti_S.tiFlag := '1'B;
    // Step 6. Receive RELEASE COMPLETE message
    U_Dc.receive (car_UplinkDirectTransfer (utran_CellDedicated,
                                            tsc_RB3,
                                            cr_U_ReleaseCompleteFacility (v_Ti_S, cr_FacilityIE_tlv))) //(cr_Components(cr_ReturnResult(cr_ResultLocNotification(tsc_InvokeId, permissionGranted))))
    -> value v_RRC_DataInd;
    v_Octetstring := v_RRC_DataInd.msg.releaseComplete.facility.components;
    v_Components := f_UTRAN_FacilityIE_Components_DecodeAndMatch(v_Octetstring, cr_Components(cr_ReturnResult(cr_ResultLocNotification(tsc_InvokeId, permissionGranted))));
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.2.3.3 Step 8");
    f_UTRAN_TestBody_Set(false);
    f_UTRAN_ReleaseCell ( utran_Cell5, f_UTRAN_CellInfo_GetConfigType (utran_Cell5) );
  }
... 
After change

...

module UTRAN_AGNSS_Testcases
{
//=========================================================================
// Imports
//=========================================================================
  import from UTRAN_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from UTRAN_ASP_Definitions language "ASN.1:2002" all;
  import from LCS_Definitions_Arguments language "ASN.1:2002" all with {encode "BER_EncRule"};
  import from TCAP_Definitions language "ASN.1:2002" all with {encode "BER_EncRule"}; //@sic R5-140685 sic@
  import from TCAP_Definitions_Para language "ASN.1:2002" all with {encode "DIRECT_EncRule"};
  import from UTRAN_Component all;
  import from UTRAN_CellInfo all;
  import from UTRAN_POS_Component all;
  import from UTRAN_ConfigurationSteps all;
  import from UTRAN_CommonFunctions all;
  import from UTRAN_TypeDefs all;
  import from UTRAN_CommonDefs all;
  import from UTRAN_AGNSS_Templates all;
  import from UpperTesterFunctions all;
  import from Parameters all;
  import from UTRAN_TTCN_ASP_Templates all;
  import from UTRAN_NAS_PDU_Templates all;
  import from NAS_24008Templates_IRAT all;
  import from UTRAN_CommonProcedures all;
  import from NAS_CommonTemplates all;
  import from UTRAN_AGNSS_Functions all;
  import from POS_Parameters all;
  import from CommonDefs all;
  import from PosSystem_CommonUTRAN all;
  import from ComponentsType all; //@sic R5-140685 sic@
  import from LCS_Objects_Templates all;
  import from PosSystem_Common all;
  import from SS_EncDecComponents all;
  import from NAS_CommonTypeDefs all;
  import from UTRAN_SS_CommonTemplates all; //@sic R5-140685 sic@
  import from SS_CommonTemplates all; //@sic R5-140685 sic@
  import from UpperTesterCommon all;
//==============================================================================
// NI-LR Emergency Call: UE-Based A-GNSS
//==============================================================================
...

//==============================================================================
// Location Notification
//==============================================================================
   /*
   * @desc      REFERENCE TS 37.571-2 clause 6.2.3.3
   * @status    NOT APPROVED
   */
   function f_TC_6_2_3_3_UTRAN() runs on UTRAN_PTC
   {
    var START_Value v_START_Value := '00000000000000000000'B;
    var RRC_DATA_IND v_RRC_DataInd;
    var TI v_Ti_S;
    var octetstring v_Octetstring;
    var octetstring v_EncodedComponents;
    var Components v_Components;
    var boolean v_RelCmpl := false;
    var boolean v_DisplayLcs := false;
    f_UTRAN_Init(UTRAN_Only);
    f_UTRAN_SS_CreateCellDCH (utran_Cell5);
    f_UTRAN_SendDefSysInfo (utran_Cell5);
    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationServingCell);
    //  Switch on UE
    f_UT_SwitchOnUE(UT, false);
    f_UTRAN_IdleUpdated (utran_Cell5);
    f_UTRAN_PageCS(utran_Cell5, terminatingConversationalCall, oct2bit(px_TMSI_Def));
    f_UTRAN_RRC_ConnEst (utran_Cell5, terminatingConversationalCall);
    U_Dc.receive (car_InitDirectTransfer (utran_CellDedicated,
                                          tsc_RB3,
                                          cr_U_PagRsp (?, cr_MobileIdAny(omit))))-> value v_RRC_DataInd;
    v_START_Value := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value);
    f_UTRAN_TestBody_Set(true);
    // Steps 1 and 2. MM Authentication
    f_UTRAN_MM_Authentication ();
    // Step 3. Security Mode procedure
    f_UTRAN_RRC_Security(utran_Cell5, true, cs_domain);
    // Step 4. Send REGISTER
    v_EncodedComponents := f_SS_EncodeComponents (cs_Components (cs_InvokeComponent (cs_InvokeLocNotification (tsc_InvokeId,
                                                                                                            notifyLocationAllowed,
                                                                                                            f_POS_Get_LcsClientAddress (px_LcsClientAddressTOA,
                                                                                                                                    px_LcsClientAddressNPI,
                                                                                                                                    px_LcsClientAddressDigits),
                                                                                                            f_CharPacking_IA5toGsm7Bit (px_LcsClientName, USSD_Packing)))));
    U_Dc.send (cas_DataReq_CS (utran_CellDedicated,
                               tsc_RB3,
                               cs_U_RegisterMT (cs_TI_MT, cs_FacilityIE_tlv(v_EncodedComponents))));
    // Steps 5 and 6 can happend in any order
    v_Ti_S := valueof(cs_TI_MT);
    v_Ti_S.tiFlag := '1'B;
    // Step 5. UE notifies the UE user of the location request
    UT.send(cas_UT_Req(CHECK_LCS_CLIENT, CNF_REQUIRED, omit));
    interleave {
      // expect RELEASE COMPLETE
      []U_Dc.receive (car_UplinkDirectTransfer (utran_CellDedicated,
                                                tsc_RB3,
                                                cr_U_ReleaseCompleteFacility (v_Ti_S, cr_FacilityIE_tlv))) -> value v_RRC_DataInd
        {
          v_RelCmpl := true;
        }
      []UT.receive(car_UT_COMMON_CNF)
        {
          v_DisplayLcs := true;
        }
    }
    //@siclog "Step 5" siclog@
    //UE notifies the UE user of the location request
    if (v_DisplayLcs) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.2.3.3 Step 5");
    }
    // Step 6. Receive RELEASE COMPLETE message
    //Receive RELEASE COMPLETE message
    if (v_RelCmpl) {
      v_Octetstring := v_RRC_DataInd.msg.releaseComplete.facility.components;
      v_Components := f_UTRAN_FacilityIE_Components_DecodeAndMatch(v_Octetstring, cr_Components(cr_ReturnResult(cr_ResultLocNotification(tsc_InvokeId, permissionGranted))));
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.2.3.3 Step 8");
    }
    f_UTRAN_TestBody_Set(false);
    f_UTRAN_ReleaseCell ( utran_Cell5, f_UTRAN_CellInfo_GetConfigType (utran_Cell5) );
  }
... 
5. Execution Log Files

5.1 Nvidia T4i UE 

The Nvidia T4i UE passed this test case on R&S CMW500 with integrity (‘uia1’) and ciphering (‘uea1’) algorithm in UTRAN FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
Rohde-Schwarz\tc_6_2_3_3_UTRAN_UE_POS_Nvidia.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
References

	[1]
	R5s140317:    Supporting information for agreement of UTRAN UE positioning TC 6.2.3.3. 
                        This archive comprises:

                         -  text format execution log files.

	
	


